

2.9 MÓDULO DE INTERRUPCIÓN EXTERNA – IRQ1 USO DEL PIN IRQ1 PARA RECONOCIMIENTO DE SEÑALES EXTERNAS

Preparado por: Rangel Alvarado
Estudiante Graduando de Lic. en Ing. Electromecánica
Universidad Tecnológica de Panamá
Panamá, Panamá
“e-mail”: issaiass@cwpanama.net
“web site”: <http://www.geocities.com/issaiass/>

ÍNDICE

2.9.1	Introducción	285
2.9.2	Registros del Módulo de Interrupción por Externa	286
2.9.3	Diagrama de Flujo	288
2.9.4	Código	289
2.9.5	Conclusión	294
2.9.6	Referencias	294
2.9.7	Problemas Propuestos	294

2.9.1 Introducción

El módulo de interrupciones externa, es un pin de múltiples funciones en estos microcontroladores, pero en el modo de aplicación en especial puede, por ejemplo, ser un dispositivo esclavo que responda a un dispositivo maestro.

El módulo de interrupciones externa proporciona:

- Un pin dedicado a interrupciones (IRQ1)
- Bits de control del registro
- Buffer con histéresis
- Reconocimiento automático de la interrupción y “pull-up” seleccionable.

Para demostrar su uso, solo se programará el pin IRQ1 para realizar una función sencilla de encendido y apagado, por consiguiente, la nota abarca:

- Configurar el pin de interrupción externa: Se configura el módulo para detectar el flanco de bajada¹.
- Generar una interrupción: La rutina demuestra que posibilidades a eliminar errores a ruido pueden tomarse, a pesar de que el módulo es autosuficiente y automático.
- Simular: Como este pin necesita alto voltaje para entrada al modo de comunicación con la PC “no” se podrá simular la aplicación.

Nota: Se asume un dominio total o parcial de la NT0108 y puede se haga referencia a esta.

¹ Solo detecta flancos de bajada

2.9.2 Registros del Módulo de Interrupción Externa

2.9.2.1 Registro de Estado y Control de la Interrupción Externa

El módulo de interrupción externa genera una atención si detecta un flanco de bajada; la señal de interrupción es borrada si: (1) se atiende la interrupción ó (2) si el pin regresa a su estado lógico.





Figura 166 (superior). Registro de Estado y Control del Módulo de Interrupción Externa. Al igual que el módulo de teclado, el módulo externo detecta solo flancos de bajada de una señal.

2.9.2.2 Registro de Configuraciones 2

1 = Resistor a cinco voltios ("pull-up"), desconectado.
 0 = "Pull-up" conectado.



Nota: El registro mostrado, es específico de los microcontroladores JK1/JK3/JL3. Los registros en los microcontroladores QT/QY y GP32 son ligeramente diferentes y para el último no se controla el estado del pin IRQ1.

Figura 167. Registro de Configuraciones 2. El bit IRQPUD deshabilita o habilita un resistor interno el cual lleva el pin a el estado de uno (1) lógico.

2.9.2.3 Interrupción del Módulo Externo IRQ1

Si las interrupciones externas no han sido enmascaradas (IMASK1), se generará una interrupción cada vez que se detecte un flanco de bajada o bajos niveles dependiendo del bit MODE1.

Tabla 64. Vectór de Interrupción del Módulo Externo

Bandera	Máscara	Dirección	Vector (Dirección)
KEY1	IMASK1	FFFA	Vector IRQ (Alto)
		FFFB	Vector IRQ (Bajo)

El vector de interrupción es uno de los más prioritarios.

R = Reservado
 = No implementado

2.9.2.4 Inicialización de Módulo de Interrupción Externa

Inicialice el módulo de interrupción externa para reconocer solo bordes de bajada y deshabilite el “pull-up” interno del dispositivo.




Figura 168. Pasos de Inicialización del Módulo de Interrupción Externa. Evite posibles falsas interrupciones siguiendo el procedimiento de inicialización de teclado. Nota: El registro CONFIG2 mostrado, es exclusivo de los microcontroladores JK1/JK3 y JL3; QT/QY y GP32 son ligeramente diferentes.

2.9.3 Diagrama de Flujo

El programa conmuta un LED en la tarjeta si un flanco de bajada es detectado por el pin IRQ1 el cual espera un tiempo de rebote.




Figura 169 (inferior-derecha). NT0109 – IRQ1. (a) Programa Principal. Inicializa el módulo de interrupción externa, el pin 1, IRQ1 es inicializado para recibir flancos de bajada. (b) Reinicio del Sistema. Al presionar “Reset”, el sistema es reiniciado sin importar su estado actual. (c) Interrupción Externa. Conmuta el estado del LED PTD7 y retarda doscientos milisegundos (200 ms) de eliminación de rebote. (d) Inicialización de Puertos. Puerto D7 es una salida en estado alto. (e) Inicialización de Módulo de Interrupción Externa. Inicializa el módulo para eliminar falsas interrupciones. (f) Retardo. Subrutina de retardo programable para eliminación de rebotes.

2.9.4 Código

```

=====
; ARCHIVO      : NT0109 - IRQ1 - 13 07 04.asm
; PROPÓSITO   : Generar una interrupcion cada vez que se presione
;              solamente la tecla conectada a IRQ1 y conmuta el LED de la
;              tarjeta conectado al puerto PTD7.
;
; NOTA        :
;              1 - Observar la rutina de inicialización del módulo externo.
;              2 - Añadir interrupciones de IRQ1.
; REFERENCIA:
;              Advanced Information of MCU68HC908JK1, JK3, JL3...
;              http://www.freescale.com/files/microcontrollers/
;              doc/data_sheet/MC68HC08JL3.pdf
;              Pág. 149 @ 153 - Módulo de interrupción Externa
;
; LENGUAJE    : IN-LINE ASSEMBLER
;
-----
; HISTORIAL
; DD MM AA
; 26 05 03 Creado.
; 10 09 04 Modificado.
=====

=====
;
; Definiciones del Usuario
=====
COPD      equ 0T           ; Bit 0 del registro CONFIG1
PTD7      equ 7T           ; Puerto D, Bit 7
DDRD7     equ 7T           ; Registro de Direccionamiento D, Bit 7
ACK1      equ %00000100   ; INTSCR, Bit de Reconocimiento de
                          ; Interrupción, Bit 2 ON
IMASK1    equ %00000010   ; INTSCR, Bit de Máscara del Registro
                          ; KBSCR, Bit 1 ON
MODE1     equ %00000001   ; INTSCR, Bit de Selección de Detección, Bit 0
                          ; ON
IRQPUD    equ %10000000   ; CONFIG2, Bit Selector de pull-up de IRQ1,
                          ; Bit 7 ON
DELAY49152 equ $0096      ; Constante de Retardo
MS200     equ $00C8       ; Constante de 200 ms de retardo

=====
;
; Mapa de Memoria del Microcontrolador
=====
;
; Registro de E/S
=====
PORTD     equ $0003       ; Dirección, Puerto D
DDRD      equ $0007       ; Registro de Direccionamiento, Puerto D

```

```

=====
;
;           Registro de Interrupción externa
;
=====
INTSCR      equ $001D           ; Registro de Estado y Control de Int. Externa

;
;           Registro de Configuraciones
;
=====
CONFIG2     equ $001E           ; Vectores de Configuración 2
CONFIG1     equ $001F           ; Vectores de configuración 1

;
;           Memoria FLASH
;
=====
FLASH_START equ $EC00           ; Puntero - Mem.FLASH

;
;           Vectores de Usuario
;
=====
IRQ1H      equ $FFFA           ; Vector de Interrupción Externa
RESET_VEC  equ $FFFE           ; Puntero del RESET

;
; OBJETIVO   : Inicio de Codif. del Ensam-
;             blador en Memoria FLASH.
;
=====
org FLASH_START           ; Inicio Mem. FLASH

;
; OBJETIVO   : Genera una interrupción de
;             tecla cada vez que se pre-
;             sione uno de los dos botones
;             configurados, si existe to-
;             davía una tecla en bajo o
;             ambas, no llama la
;             interrupción.
;
=====
START
    rsp                ; Inic.Stack = $00ff
    bset COPD,CONFIG1  ; Desactiva watchdog
    clra               ; Borra A
    clrx               ; Borra X
    jsr INIT_PORTS     ; Inicializa Puertos
    jsr INIT_IRQ1      ; Inicializa Módulo de Int. Externa
    cli                ; Habilita Interrupciones

ESPERA
    wait               ; Espera la interrupción.
    bra ESPERA         ; Salta al modo de bajo consumo

```

```
=====
; INIT_PORTS : Inicializa variables y regis
;             tros.
; OBJETIVO   : Inicializa los registros de
;             direccionamiento.
;             PORTD7 = HIGH
;             PORTD7 = OUTPUT
; ENTRADA    : Ninguna
; SALIDA     : Ninguna
; REGISTROS
; AFECTADOS  : PORTD, DDRD
=====
INIT_PORTS
    bset PTD7,PORTD          ; Puerto D 7 en alto
    bset DDRD7,DDRD         ; Puerto D 7 es salida
    rts                     ; retorna

=====
; INIT_IRQ1  : Inicializa el módulo de inte-
;             rrupciones externa.
; OBJETIVO   : Tecla en IRQ1 responde a la
;             presión de ella.
; ENTRADA    : Ninguna
; SALIDA     : Ninguna
; REGISTROS
; AFECTADOS  : INTSCR
=====
INIT_IRQ1
    mov #IMASK1,INTSCR      ; Bajos niveles y bordes
    mov #IRQPUD,CONFIG2    ; Deshabilita pull-up
    mov #ACK1,INTSCR       ; Borro la falsa interrupción si existe
    rts                    ; Retorno de la subrutina.
```

```

;=====
; DELAY      : Genera un retardo de tiempo
; OBJETIVO   : Retardo de tiempo, base 1ms
; ENTRADA    : H:X = Retardo en ms
; SALIDA     : H:X = 0
; REGISTROS
; AFECTADOS  : H:X
; USO
;           :
;           : MIN = H:X = 1T
;           : MÁX = H:X = 65535T
;           : ldhx #500
;           : jsr Delay ; retarda 0.5 seg
;=====
Delay pshx                ; [2] Salva X en la pila
      pshh                ; [2] Salva H en la pila
      ldhx #DELAY49152    ; [3] Carga constante de bucle fino
Delay0 aix #-1           ; [2] Decrementa H:X en 1
      cphx #0             ; [3] Llegó a cero (0)
      bne Delay0         ; [3] Si no es igual, salta a Delay0
      pulh               ; [2] Si es igual, recupera H de la pila
      pulx               ; [2] Recupera X de la pila
      aix #-1           ; [2] Decrementa H:X en 1
      cphx #0           ; [3] Llegó a cero (0)
      bne Delay         ; [3] Si no es igual, salta a Delay
      rts               ; [4] retorna

;=====
; IRQ1L     : Interrupción Externa
; OBJETIVO   : Reconoce la interrupción al
;           : presionarse la tecla
; ENTRADA    : Ninguna
; SALIDA     : Ninguna
; REGISTROS
; AFECTADOS  : INTSCR, H:X, PORTD, A
;=====
IRQ1L
      mov #IMASK1,INTSCR ; Enmascara interrupciones
      lda PORTD          ; Lee el puerto
      coma              ; cambia 1's por 0's
      sta PORTD         ; PORTD = A
      ldhx #MS200       ; A retardar 200 ms
      jsr Delay         ; Retarda 200 ms, elimina rebotes
      mov #ACK1,INTSCR  ; Borra interrupciones subsiguientes (ACK1
                        ; = 1).
      rti               ; Retorno de la interrupción.

```



```
=====
; OBJETIVO : Inicializa el Vector de Reset
;           Arranque del programa en la
;           memoria Flash y búsqueda de la
;           interrupción externa
=====
;===== Vectores de IRQ1, Tecla a apretar =====
org IRQ1H ; Vector de Interrupción Externa
dw IRQ1L ; Tecla apretada
;===== Vector de Reinicio de Sistema =====
org RESET_VEC ; Puntero Vec - RESET
dw START ; al darse reset salta a Start
```

Listado 18. NT0109 – IRQ1. El programa responde a inicializar el módulo de interrupción externa y cada vez que se presiona, se generará una interrupción la cual conmutará el LED de la tarjeta conectado al puerto PTD7.

Nota: Este código no se podrá simular; el pin IRQ1 es necesario para la entrada al modo de comunicación con la PC. Queme su pastilla según la nota NT0009, sección 1.9.5, remueva el jumper de Programación/Aplicación y pulse el interruptor conectado al pin IRQ1 para visualizar su función.

2.9.5 Conclusión

El pin de interrupción externa es una alternativa más para adquirir una señal de referencia externa, o también si queremos una comunicación de una sola dirección con un dispositivo maestro.

Además, el pin IRQ1 es un pin multifunción (el microcontrolador necesita ver un voltaje alto [8.2 V] para programar la pastilla sobre este pin) comprometido y por este motivo la nota no se pudo simular, pero el método de operación es muy parecido al módulo de teclado, lo que facilita su entendimiento.

2.9.6 Referencias

2.9.6.1 Información Avanzada sobre el Microcontrolador

(a) http://www.freescale.com/files/microcontrollers/doc/data_sheet/MC68HC08JL3.pdf

Págs. 149 a 153 – Módulo de Interrupción Externa (IRQ1).

2.9.6.2 Manual de Referencia del CPU

(a) http://www.freescale.com/files/microcontrollers/doc/ref_manual/CPU08RM.pdf

Pág. 114 – Instrucción BIH – Salta si el pin IRQ1 está en estado alto.

Pág. 115 – Instrucción BIL – Salta si el pin IRQ1 está en estado bajo.

2.9.6.3 Página “web” sobre esta Nota Técnica

(a) <http://www.geocities.com/issaiass/>

2.9.7 Problemas Propuestos

2.9.7.1 Genere un contador que después de cinco (5) pulsaciones encienda un LED.

2.9.7.2 Utilice el módulo de interrupción externa y detecte por cada pulsación una secuencia diferente para el LED cada vez que toca un interruptor.

2.9.7.3 Genere un eliminador de rebotes con el temporizador para una tecla conectada al pin IRQ1.